117 CS 538 Spring 2002^©

Recursion in Scheme

Recursion is widely used in Scheme and other functional programming languages. Rather than using a loop to step through the elements of a list or array,recursionbreaksa problemona large data structure into a simpler problem on a smaller data structure. A good example of this approach is the append function, which joins (or appends)twolistsintoonelargerlist containing all the elements of the two input lists (in the correct order). Note thatcons is notappend. cons adds one element to the head of an existing list.

118 CS 538 Spring 2002^©

Thus

(cons '(a b) '(c d)) ⇒
   ((a b) c d)
(append '(a b) '(c d)) ⇒
   (a b c d)

Theappend function is predefined in Scheme, as are many other useful list-manipulating functions (consult the Scheme definition for what’s available). It is instructive to defineappend directly to see its recursive approach:

(define
  (append L1 L2)
  (if  (null? L1)
        L2
        (cons (car L1)
              (append (cdr L1) L2))
  )
)

119 CS 538 Spring 2002^©

Let’s trace(append '(a b) '(c d)) Our definition is

(define
  (append L1 L2)
  (if  (null? L1)
        L2
        (cons (car L1)
              (append (cdr L1) L2))
  )
)

NowL1 = (a b) and L2 = (c d).

(null? L1) is false, so we evaluate

(cons (car L1) (append (cdr L1) L2))
= (cons (car '(a b))
       (append (cdr '(a b)) '(c d)))
= (cons 'a (append '(b) '(c d))

We need to evaluate

 (append '(b) '(c d))

Inthiscall,L1 = (b) andL2 = (c d).

L1 is not null, so we evaluate

(cons (car L1) (append (cdr L1) L2))
= (cons (car '(b))
        (append (cdr '(b)) '(c d)))

120 CS 538 Spring 2002^©

= (cons 'b (append '() '(c d))

We need to evaluate

 (append '() '(c d))

In this call,L1 = () and L2 = (c d).

L1 is null, so we return(c d).

Therefore

(cons 'b (append '() '(c d)) =
(cons 'b '(c d)) = (b c d) =
(append '(b) '(c d))

Finally,

(append '(a b) '(c d)) =
(cons 'a (append '(b) '(c d)) =
(cons 'a '(b c d)) = (a b c d)

Note: Source files forappend, and other Scheme examples, may be found in

~cs538-1/public/scheme/example1.scm,
~cs538-1/public/scheme/example2.scm,
etc.

121 CS 538 Spring 2002^©

Reversing a List

Another useful list-manipulation function isrev, which reverses the members of a list. That is, the last element becomes the first element, thenext-to-lastelementbecomesthe second element, etc. For example,

(rev '(1 2 3)) ⇒  (3 2 1)

The definition ofrev is straightforward:

(define (rev L)
  (if (null? L)
      L
      (append (rev (cdr L))
              (list (car L))
      )
  )
)

122 CS 538 Spring 2002^©

As an example, consider

(rev '(1 2))

HereL = (1 2). L is not null so we evaluate

(append (rev (cdr L))
              (list (car L))) =
(append (rev (cdr '(1 2)))
              (list (car '(1 2)))) =
(append (rev '(2)) (list 1)) =
(append (rev '(2)) '(1))

We must evaluate(rev '(2)) HereL = (2). L is not null so we evaluate

(append (rev (cdr L))
              (list (car L))) =
(append (rev (cdr '(2)))
              (list (car '(2)))) =
(append (rev ())(list 2)) =
(append (rev ())'(2))

We must evaluate(rev '()) HereL = (). L isnullso(rev '())=

()

123 CS 538 Spring 2002^©

Thus(append (rev ())'(2)) =

(append  () '(2)) = (2) = (rev '(2))

Finally, recall(rev '(1 2)) =

(append (rev '(2)) '(1)) =
(append '(2) '(1)) = (2 1)

As constructed, rev only reverses the “top level” elements of a list. That is, membersofa listthat themselvesare listsaren’t reversed. For example,

 (rev  '( (1 2)  (3 4))) =
 ((3 4) (1 2))

We cangeneralizerev to alsoreverse list members that happen to be lists. Todothis,it willbeconveniento use Scheme’s let construct.

124 CS 538 Spring 2002^©

The Let Construct

Scheme allows us to create local names, boundto values, foruseinan expression. The structure is

(let ( (id1 val1) (id2 val2) ... )
  expr )

In this construct,val1 is evaluated and bound toid1, which will exist only within thislet expression. If id1 is already defined (as a global or parameter name) the existing definition is hidden and the local definition, bound toval1, is used. Thenval2 is evaluated and bound to id2,....Finally, expr isevaluatedina scope that includesid1, id2, ...

125 CS 538 Spring 2002^©

For example,

(let  ( (a 10)  (b 20) )
  (+ a b)) ⇒ 30

Using alet, the definition of

revall, a version ofrev that

reverses all levels of a list, is easy:

(define (revall L)
  (if (null? L)
      L
      (let ((E (if (list? (car L))
                 (revall (car L))
                 (car L) )))
        (append (revall (cdr L))
                (list E))
      )
  )
)

(revall  '( (1 2)  (3 4))) ⇒
 ((4 3) (2 1))

126 CS 538 Spring 2002^©

Subsets

Another good example of Scheme’s recursive style of programming is subset computation. Given a list of distinct atoms, we want to compute a list of all subsets of the list values. For example,

 (subsets  '(1 2 3)) ⇒
  ( () (1) (2) (3) (1 2) (1 3)
   (2 3) (1 2 3))

The order of atoms and sublists is unimportant, but all possible subsets of the list values must be included. Given Scheme’s recursive style of programming, we need a recursive definition of subsets.

127 CS 538 Spring 2002^©

Thatis, if wehavea listofallsubsets of n atoms, how do we extend this list to one containing all subsets of n+1 values? First, we note that the number of subsetsofn+1valuesisexactlytwice the number of subsets of n values. For example,

(subsets '(1 2) ) ⇒
( () (1) (2) (1 2)), which

contains 4 subsets.

(subsets '(1 2 3)) contains 8

subsets (as we saw earlier). Moreover, the extended list (of subsets for n+1 values) is simply the list of subsets for n valuesplus the resultof “distributing”the newvalue into each of the original subsets.

128 CS 538 Spring 2002^©

Thus(subsets '(1 2 3)) ⇒

( () (1) (2) (3) (1 2) (1 3)
  (2 3) (1 2 3)) =
( ()   (1)   (2)   (1 2) ) plus
( (3) (1 3) (2 3) (1 2 3) )

This insight leads to a concise program for subsets. We will let(distrib L E) be a function that “distributes”E into each list inL. For example,

(distrib '(() (1) (2) (1 2)) 3) =
( (3) (3 1) (3 2) (3 1 2) )
(define (distrib L E)
  (if (null? L)
      ()
      (cons (cons E (car L))
            (distrib (cdr L) E))
   )
)

129 CS 538 Spring 2002^©

We will let(extend L E) extend a listL by distributing elementE throughL and then appending this result toL. For example,

 (extend '( () (a) ) 'b) ⇒
 ( () (a) (b) (b a))

(define (extend L E)
  (append L (distrib L E))
)

Nowsubsets is easy:

(define (subsets L)
  (if (null? L)
      (list ())
      (extend (subsets (cdr L))
              (car L))
  )
)